<?xml version="1.0" encoding="windows-1250"?>
<gxsl:stylesheet xmlns:gxsl="http://www.w3.org/1999/XSL/Transform" version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/TransformAlias" xml:space="default">
  <gxsl:namespace-alias stylesheet-prefix="xsl" result-prefix="gxsl"/>
  <gxsl:output method="xml" encoding="windows-1250" indent="yes"/>
  
  <gxsl:param name="cvincludepage"/>
  <gxsl:param name="lang">cs</gxsl:param>
  
  <gxsl:template match="/">
    <xsl:stylesheet version="1.0">
      <xsl:output method="html" indent="no" encoding="windows-1250"/>
      <gxsl:apply-templates select="$cvincludepage/ancestor::CView"/>
    </xsl:stylesheet>
  </gxsl:template>

  <gxsl:template match="CView">

    <gxsl:variable name="dict" select="document('dictionary.xml')/Translation"/>

    <xsl:template match="DPEdit"/>

    <xsl:template match="coverer_{CVInclude/@concept}|coverer_{CVInclude/@concept}_norec">      
      <form action="?dp=createxml" method="post">
        <!--<textarea class="text" cols="110" rows="18" name="indoc"><xsl:copy-of select="."/></textarea><br/>-->
        <textarea class="text" cols="110" rows="18" name="indoc"><xsl:apply-templates select="." mode="escape"/></textarea><br/>        
        <xsl:if test="/DPOutput/@access!='r'">
          <gxsl:variable name="title"><xsl:value-of select="CVInclude/CVITitles/CVITitle[@id_cttype='createbutton']/Title"/></gxsl:variable>
          <gxsl:choose>
            <gxsl:when test="not($title) or $title=''">
              <input type="submit" class="input_button"><gxsl:attribute name="value"><gxsl:value-of select="$dict/Text[@tid='crt']/Value[@lang=$lang]"/></gxsl:attribute></input>
            </gxsl:when>
            <gxsl:otherwise>
              <input type="submit" class="input_button" value="{$title}"/>
            </gxsl:otherwise>
          </gxsl:choose>                  
        </xsl:if>
      </form>      
    </xsl:template>

    <xsl:template match="@*" mode="escape">
  		<xsl:attribute name="{{name()}}">
    		<xsl:call-template name="do_escaping">
    			<xsl:with-param select="." name="text"/>
    		</xsl:call-template>
      </xsl:attribute>        		
    </xsl:template>

  	<xsl:template match="node()" mode="escape">
   		<xsl:copy>
    	  <xsl:choose>
    	    <xsl:when test="text()">
        		<xsl:call-template name="do_escaping">
        			<xsl:with-param select="." name="text"/>
        		</xsl:call-template>
    	    </xsl:when>
    	    <xsl:otherwise>
      			<xsl:apply-templates select="@*|node()" mode="escape"/>
    	    </xsl:otherwise>
    	  </xsl:choose>
  		</xsl:copy>
  	</xsl:template>
	
  	<xsl:template name="do_escaping">
  		<xsl:param name="text"/>
  		<xsl:choose>
  			<xsl:when test="contains($text,'&lt;')">			
  				<xsl:value-of select="substring-before($text,'&lt;')"/>&amp;lt;<xsl:call-template name="do_escaping">
  					<xsl:with-param select="substring-after($text,'&lt;')" name="text"/>
  				</xsl:call-template>
  			</xsl:when>
  			<xsl:when test="contains($text,'&gt;')">			
  				<xsl:value-of select="substring-before($text,'&gt;')"/>&amp;gt;<xsl:call-template name="do_escaping">
  					<xsl:with-param select="substring-after($text,'&gt;')" name="text"/>
  				</xsl:call-template>
  			</xsl:when>
  			<xsl:when test="contains($text,'&amp;')">			
  				<xsl:value-of select="substring-before($text,'&amp;')"/>&amp;amp;<xsl:call-template name="do_escaping">
  					<xsl:with-param select="substring-after($text,'&amp;')" name="text"/>
  				</xsl:call-template>
  			</xsl:when>
  			<xsl:when test="contains($text,'&quot;')">			
  				<xsl:value-of select="substring-before($text,'&quot;')"/>&amp;quot;<xsl:call-template name="do_escaping">
  					<xsl:with-param select="substring-after($text,'&quot;')" name="text"/>
  				</xsl:call-template>
  			</xsl:when>
  			<xsl:otherwise>
  				<xsl:value-of select="$text"/>
  			</xsl:otherwise>
  		</xsl:choose>
  	</xsl:template>

    <xsl:template name="title"><gxsl:value-of select="ancestor::CView/CVTitles/CVTitle/Title"/></xsl:template>  

  </gxsl:template>

	
</gxsl:stylesheet>